
Се год ня мы будем ломать мой crackme, напич канный анти отла доч ными при -
ема ми. Они осно ваны на осо бен ностях обра бот ки исклю чений отладчи ками
и на ошиб ках в  debug engine, о которых я рас ска жу по ходу дела. А заод но
про демонс три рую интимные под робнос ти основных хакер ских инс тру мен -
тов — «Оль ги», «Иды», Syser’а, x86emu и про чих.

Раз гадывать загад ки нам ного инте рес нее, чем читать готовые решения. А  потому, пока
еще  не поз дно, отор вись от  статьи и  поп робуй рас ковырять JohoR crackme. В  под сказ ку
не заг лядывать! Исходный текст не смот реть! Впро чем, сам по себе исходный текст (даже
с уче том всех ком мента риев) совер шенно не объ ясня ет, как же его отла живать.

WWW

;  так же есть в моем репози тории
на OpenRCE.
Ска чать crackme-jhr ко пия

Здесь отсутс тву ют шиф ровка, самомо дифи кация и про чие при емы, ослепля ющие ста тичес -
кий ана лиз. Дизас сем блер выда ет акку рат ный лис тинг, каж дая машин ная коман да которо го
абсо лют но понят на. Одна ко резуль тат дей ствия прог раммы в целом очень труд но пред ска -
зуем и тре бует доволь но глу боких зна ний устрой ства про цес сора и опе раци онной сис темы.
Поис тине тан таловы муки! Какой‑то нес час тный десяток машин ных инс трук ций (ядро
crackme) отде ляет нас от победы! Что ж, тем боль шее нас лажде ние испы тыва ешь от взло ма!
Ну а на слу чай, если самому взло мать никак не получа ется, я даю раз верну тое объ ясне ние.

 

ЧТО МЫ БУДЕМ ЛОМАТЬ

Ис ходный текст JohoR crackme при веден в лис тинге  1. Это  чудо моей инже нер ной мыс ли
пос ле ком пиляции занима ет все го  832  бай та, боль шая часть которых при ходит ся на  PE-
заголо вок. Конеч но, его мож но было бы ужать, прог рамми руя в hex-кодах, но это ж сколь ко
тру да надо пот ратить! А так — фай лы лег ко ком пилиру ются штат ными ути лита ми Microsoft.

Syser нап рочь отка зыва ется гру зить JohoR crackme в release build’е

Кста ти, о ком пиляции. По мно гочис ленным прось бам тру дящих ся я отка зал ся от коман дных
фай лов и перешел на макак (в смыс ле, на  .mak), обра баты ваемых ути литой NMAKE из ком -
плек та пос тавки MS Visual Studio.  собира ет релиз, а 

 — отла доч ную вер сию. Толь ко все
рав но отла дить ее с помощью MS Visual Studio не удас тся — нет смыс ла даже пытать ся.

NMAKE /f crackme_jhr.mak NMAKE /f

"crackme_jhr.mak" CFG="crackme_jhr - Win32 Debug"

Так же под держи вает ся сбор ка и из IDE — дос таточ но открыть макаку и сде лать build. Тупая
«Сту дия» всег да ищет ском пилиро ван ный файл в  катало гах  и  , тог да
как  мыщъх соз дает его в  текущей дирек тории, поэто му запуск фай ла непос редс твен но
из IDE невоз можен (хотя, может быть, в пос ледних вер сиях MS уже пофик сила этот косяк).

\Debug \Release

Ис ходный текст JohoR crackme

#include <windows.h>
   int count; char str[]="0123456789ABCDEF!";

 __declspec(naked) nezumi()
{
    __asm{
     ; //int 03  ; // For soft-ice
        xor eax, eax ; // eax := 0
          mov ebx, fs:[eax] ; // Old SEH
      pushfd ; // save EFLAGS
    ;//-[new seh]-;
         push offset l1 ; // Handler proc
       push -1 ; // The last handler in the chain
        mov fs:[eax], esp ; // Assign the new handler
    ;//-[hacker time]-;
         xor eax,[eax] ; // <-- ACCESS VIOLATION
    ;//-[set TF bit]-;
       push -1 ; // TF := 1
         xor eax,[eax] ; // <-- ACCESS VIOLATION
         popfd ; // EFLAGS := 00244ED7h
    ;//-[TRACE-ZONE]-;
         mov eax, [eax] ; // <-- ACCESS VIOLATION
      nop ; // <-- INT 01
      ud2 ; // <-- ILLEGAL INSTRUCTION
      nop ; // <-- INT 01
      nop ; // <-- INT 01
       int 03 ; // <-- INT 01
       jmp end_of_line ; // :-- to exit -->
    ;//-[seh handler]- -;
           l1: mov eax, [esp + 04h] ; // *EXCEPTION_RECORD
           l2: mov edx, [esp + 0Ch] ; // EDX -> ContextRecord
         mov eax, [eax] ; // EXCEPTION CODE
        cmp eax, 0C000001Dh ; // ILLEGAL INSTRUCTION
          jz x2 ; // X-->
        cmp eax, 080000003h ; // INT 03
         jz x1 ; // -- skip 1 byte -->
        cmp eax, 0C0000005h ; // ACCESS VIOLATION
        jnz set_tf_bit ; // -- don’t skip -->
         x2:inc dword ptr [edx+0B8h] ; // Skip one byte
         x1:inc dword ptr [edx+0B8h] ; // Skip one byte
       set_tf_bit: ; // <--X
           cmp dword ptr [edx + 0B8h], offset end_of_line
       jae end_of_handler ; // dont set TF-bit _outside_ trace-zone
            or dword ptr [edx+0C0h],100h ; // <---- set TF-bit _inside_ trace-zone
        end_of_handler:xor eax,eax ; // EXCEPTION_CONTINUE_SEARCH
        inc [count] ; // EXCEPTION COUNT
      ret ; // end of the handler
    ;//-[exit]-;
          end_of_line:mov fs:[eax],ebx ; // Restore the old SEH
         sub esp, 8 ; // restore the stack
        popfd ; // restore the flags
    }
    // Print EXCEPTION COUNT
       count = str[(count>0x10)?0x10:count]; MessageBox(0,&count,"JohoR",MB_OK);
    ExitProcess(0);
}

По пыт ка отладки бил да JohoR crackme под Syser’ом. Отладчик в панике, сис тема в ауте, хакер
на изме не

 

АЛГОРИТМ

Пер вые три коман ды сох раня ют ука затель на текущую (сис темную) SEH-запись в регис тре
EBX и затал кива ют в стек фла ги про цес сора, попут но обну ляя EAX. Сле дующие три коман ды
уста нав лива ют новый SEH-обра бот чик, находя щий ся по сме щению l1, замыкая SEH-цепоч ку
тер миниру ющим ука зате лем -1 ( ). Он сиг нализи рует сис теме о  том, что дан ный
обра бот чик  — пос ледний. Вот такой малень кий трюк (почему‑то боль шинс тво хакеров
добав ляют свой обра бот чик к цепоч ке сущес тву ющих — хотя переда вать им управле ние все
рав но не собира ются, зачем же тог да усложнять код?).

FFFFFFFFh

Ре зуль тат работы JohoR crackme при запус ке без отладчи ка

Сра зу же пос ле уста нов ки SEH-обра бот чика выпол няет ся коман да , «выб -
расыва ющая» исклю чение дос тупа типа ACCESS VIOLATION. Опе раци онная сис тема ловит
его и переда ет управле ние на мет ку l1, с кодом , рас положен ным в двой ном сло ве
по  адре су . Обра бот чик видит, что это  ACCESS VIOLATION, и, зная, что его
выраба тыва ет инс трук ция , лезет в регис тро вый кон текст. При этом он уве -
личи вает зна чение EIP на  два бай та  — , пос коль ку ACCESS
VIOLATION пред став ляет собой fault. Если пояс нять, то в  момент генера ции исклю чения
регистр EIP ука зыва ет на начало воз будив шей его машин ной коман ды. При выходе из обра -
бот чика про цес сор будет выпол нять  сно ва и сно ва, пока мы либо не изме -
ним EAX так, что бы он ука зывал на валид ную область памяти, либо не уве личим зна чение
EIP, перехо дя к выпол нению сле дующей машин ной коман ды. Что мы и дела ем, попут но уве -
личи вая счет чик вызова исклю чений (count) на еди ницу.

XOR EAX, [EAX]

C0000005h

[ESP + 04h]

XOR EAX, [EAX]
sizeof(XOR EAX, [EAX])

XOR EAX, [EAX]

За чем нам это? При  пошаго вой трас сиров ке прог раммы, ког да взве ден TF-бит, опе раци -
онная сис тема сле дом за  ACCESS VIOLATION генери рует SINGLE STEP. В резуль тате вмес то
одно го исклю чения мы получа ем целых два, и SEH-обра бот чик под отладчи ком вызыва ется
дваж ды, поз воляя прог рамме обна ружить, что ее лома ет злоб ный хакер! Отладчи ки MS VC,
MS WinDbg, Soft-Ice (и ряд дру гих) давят исклю чение SINGLE STEP, сбра сывая флаг трас -
сиров ки через кон текст, а вот «Оль га» 1.1x об этом не заботит ся. Ошиб ка была исправ лена
толь ко в вер сии 2.x (все еще находя щей ся в раз работ ке). Под робнее об этом мыщъх рас ска -
зыва ет в сво ем бло ге ( ).bug in Olly, Windows behavior and Peter Ferrie

Три сле дующие машин ные коман ды взво дят флаг трас сиров ки. На самом деле флаг трас -
сиров ки взво дит ся с помощью все го двух команд — / , а  , рас -
положен ная меж ду ними, встав лена для борь бы с одним экспе римен таль ным отладчи ком,
что «отлавли вает» инс трук цию POPFD. Если лома емая прог рамма взво дит бит трас сиров ки,
отладчик вру бает модуль эму ляции, но, если управле ние на POPFD переда ется пос редс твом
прав ки регис тро вого кон тек ста, отладчик это го не про сека ет. Точ нее, рань ше не про секал,
а сей час ошиб ка исправ лена.

PUSH -1 POPFD XOR EAX, [EAX]

POPFD вытал кива ет -1 ( ) из сте ка, уста нав ливая про цес сорные фла ги в еди ницу.
Конеч но, далеко не все фла ги, а толь ко те, которые поз волено модифи циро вать прик ладной
прог рамме. О  чем, кста ти говоря, «догады вает ся» далеко не  каж дый эму лиру ющий
отладчик, а потому зна чение  под «живым» про цес сором и, нап ример, x86emu силь но
отли чают ся. Но  x86emu, в общем‑то, и не под ряжал ся эму лиро вать все фла ги про цес сора.
В прин ципе, здесь мож но вста вить про вер ку — если  не рав няет ся , то одно
из двух: либо нас лома ют на эму лято ре, либо это какой‑то очень левый про цес сор. Впро чем,
пос коль ку дос той ных эму лиру ющих отладчи ков под  x86 все рав но нет, подоб ная про вер ка
лишена смыс ла.

FFFFFFFFh

EFLAGS

EFLAGS 00244ED7h

Но не будем отвле кать ся. Флаг трас сиров ки успешно взве ден, и по завер шении коман ды,
сле дующей за POPFD, про цес сор генери рует пошаго вое исклю чение. А этой коман дой явля -
ется наша ста рая зна комая , «выб расыва ющая» ACCESS VIOLATION, что
пред шес тву ет пошаго вому исклю чению. В  дан ном слу чае сис тема генери рует два впол не
закон ных исклю чения. Вот толь ко боль шинс тво отладчи ков оши боч но при нима ют исклю -
чение, сге нери рован ное прог раммой, за  свое собс твен ное и  давят его. В  резуль тате чего
SEH-обра бот чик вызыва ется на один раз мень ше. «Оль га» 1.1х эту ситу ацию обра баты вает
впол не пра виль но (точ нее, никак не  обра баты вает, тупо переда вая все исклю чения прог -
рамме), а вот исправ ленная «Оль га» 2.х пута ется в исклю чени ях и «давит» лиш нее (с ее точ -
ки зре ния) пошаго вое исклю чение. Дру гими сло вами, под  «Оль гой» 1.1х SEH-обра бот чик
вызыва ется на один раз боль ше, чем под «живым» про цес сором (с уче том пер вой коман ды

), а под «Оль гой» 2.х — на один раз мень ше. Кра сота! И какую же вер сию нам
выбирать? Что каса ется Soft-Ice (и некото рых дру гих отладчи ков), он «куша ет» все пошаго -
вые исклю чения, генери руемые отла жива емой прог раммой, обла мывая самот расси ров ку.
Потому SEH-обра бот чик вызыва ется толь ко на    — как  следс твие, счет чик
вызовов count ока зыва ется нам ного мень ше, чем ожи дает защита, сра зу же понима ющая,
с кем она име ет дело.

XOR EAX, [EAX]

MOV EAX, [EAX]

MOV EAX, [EAX]

Ко ман да NOP «чес тно» генери рует пошаго вое исклю чение (ведь бит трас сиров ки взве ден!),
но Soft-Ice его пог лоща ет. Осталь ные отладчи ки (типа «Оль ги» и IDA-Pro) хотя бы мож но нас -
тро ить на  отда чу пошаго вых исклю чений лома емой прог рамме. При чем IDA-Pro 5.2  пред -
ложит сде лать это авто мати чес ки, в то вре мя как «Оль га» тре бует руч ной нас трой ки (вклад -
ка Exceptions в опци ях отладчи ка).

Ре акция Syser’а на машин ную коман ду UD2

Инс трук ция UD2 генери рует исклю чение типа ILLEGAL INSTRUCTION, перех ватыва емое Soft-
Ice и не отда ваемое отла жива емой прог раммой вплоть до отда чи коман ды faults off. Syser
такой коман ды вооб ще не зна ет, обзы вая ее как DB («объ явить байт») и невер но дизас сем -
бли руя весь пос леду ющий код (что не покажет ся уди витель ным, если вспом нить, что UD2 —
двух бай товая коман да).

Па ра пос леду ющих NOP’ов не дела ет ничего, кро ме генера ции пошаго вого исклю чения, осо -
бен ность обра бот ки которо го мы обсужда ли дву мя абза цами выше. Так зачем же тог да они
нуж ны? Под сказ ка — раз ные отладчи ки име ют раз ные баги, «съедая» раз личное количес тво
исклю чений. Пра виль но! Дан ный crackme опре деля ет тип отладчи ка по зна чению count, уни -
каль ность которо го обес печива ется соот ношени ем команд, генери рующих свои собс твен -
ные исклю чения, к  обще му количес тву трас сиру емых инс трук ций. Если убрать NOP’ы,
crackme про дол жит детек тить активную отладку, но уже не смо жет опре делить, какой имен -
но отладчик исполь зует ся хакером.

Ре зуль тат работы JohoR crackme при запус ке под IDA-Pro 4.7

Ко ман да  так же встав лена нес прос та. Если даже нас тро ить отладчик на отда чу 
 лома емой прог рамме, наличие  сущес твен но зат рудня ет отладку. Если бы INT

03h не  было, то, что бы быс тро выб рать ся из  глу бин сис темно го обра бот чика исклю чений
назад к лома емой прог рамме, дос таточ но было  бы пок рыть трас сиру емый блок прог рам -
мны ми точ ками оста нова (в «Оль ге» для это го нуж но на каж дой коман де нажать F2). Прог -
рам мные точ ки оста нова пред став ляют собой одно бай товую инс трук цию CCh, лег ко обна -
ружи ваемую под сче том кон троль ной сум мы и  «раз валива ющую» самомо дифи циру ющий
код. Впро чем, ни того, ни дру гого в crackme нет, а есть толь ко . Чис то теоре тичес ки
отладчик может (и дол жен) отли чать свои собс твен ные  от  чужих, отда вая прог -
рамме толь ко те исклю чения, которые она сама же и  сге нери рова ла. Но  на прак тике
отладчи ки пута ются. «Оль га», нас тро енная на отда чу INT 03h лома емой прог рамме, при уста -
нов ке прог рам мной точ ки оста нова поверх  цик лится, вынуж дая хакера при менять
аппа рат ные точ ки оста нова (которых все го четыре) или точ ки оста нова на реги он памяти,
реали зован ные через под мену атри бутов стра ниц, что так же лег ко обна ружи вает ся.

INT 03h INT

03h INT 03h

INT 03h

INT 03h

INT 03h

Кста ти, с точ ки зре ния про цес сора,  генери рует trap, а не fault. То есть регистр EIP
в  момент генера ции исклю чения смот рит на  коман ду, сле дующую за  , которой
в нашем слу чае явля ется двух бай товая инс трук ция . В чем же под вох? А в
том, что SEH-обра бот чик отлавли вает BREAKPOINT-исклю чение (соот ветс тву ющее коду

) и  уве личи вает зна чение EIP на  еди ницу. Неуже ли управле ние переда ется
в середи ну инс трук ции ? Какой хит рый при ем про тив дизас сем бле ра! Гм,
вот толь ко непонят но... пер вый байт опко да JMP SHORT равен EBh, вто рой пред став ляет
отно ситель ное сме щение целево го перехо да. И  что бы оно соот ветс тво вало осмыслен ной
машин ной инс трук ции, необ ходимо, что бы мет ка  рас полага лась на опре делен -
ном сме щении от  коман ды JMP. А  в crackme меж ду ними рас положен SEH-обра бот чик.
Выходит, если его изме нить, то crackme сра зу перес танет работать? Такая хит рая защита
исходных тек стов от изме нения!

INT 03h

INT 03h
JMP END_OF_LINE

080000003h

JMP END_OF_LINE

END_OF_LINE

И чего толь ко со стра ху не покажет ся. Да, в руководс твах от Intel чер ным по белому написа -
но, что BREAKPOINT — это trap, а не fault. Вот толь ко SEH-обра бот чик вызыва ется не про цес -
сором, а  опе раци онной сис темой. Той, что написа на ком пани ей Microsoft. А  Microsoft Way
умом не понять. Ну чем мож но объ яснить, что она под меня ет про цес сорный кон текст, умыш -
ленно умень шая EIP на  еди ницу? Пар ни из  Microsoft впо пыхах забыли, что BREAKPOINT
может генери ровать ся как  опко дом CCh, так и  CDh 03h, а  потому если внед рить CDh 03h
в прог рамму и никак ее не обра баты вать, то пос ле выхода из исклю чения регистр EIP будет
смот реть на опкод  03h, соот ветс тву ющий коман де ADD чего‑то там. Допус тим, за 
сле дует ССh (еще один INT 03h, толь ко слег ка дру гой). Тог да про цес сор выпол нит опкод 

 — . Вот и поп робуй догадать ся об этом при дизас сем бли рова нии!

CDh 03h
03h

CCh ADD ECX, ESP

Ус танов ка прог рам мных точек оста нова внут ри «горячей» зоны

На конец, коман да  выводит код из зоны трас сиров ки. Прог рамма вос ста -
нав лива ет преж ний SEH, вытал кива ет из  сте ка фла ги и рас печаты вает зна чение счет чика
исклю чений, пос ле чего завер шает свое выпол нение вызовом фун кции .

JMP END_OF_LINE

ExitProcess(0)

 

СЧАСТЛИВЫЙ ФИНАЛ

Так как же все‑таки лома ют эти прог раммы? И какими отладчи ками? В слу чае ста тичес кого
кода (к которо му отно сит ся мой crackme) проб лема реша ется уста нов кой точ ки оста нова
за пре дела ми «горячей» зоны, где про исхо дит выб рос исклю чений, с про гоном их на живом
про цес соре (без пошаго вой трас сиров ки). Если же нам жиз ненно необ ходимо под смот реть
зна чение некото рых регис тров или яче ек памяти внут ри «горячей» зоны — на них уста нав -
лива ется аппа рат ная точ ка оста нова.

Ди нами чес кий код (упа кован ный, зашиф рован ный, самомо дифи циру ющий ся) заломать нам -
ного слож нее, пос коль ку нам реаль но необ ходимо прог нать его через пошаго вый трас сиров -
щик, с которым и борет ся защита. Заметим, весь ма эффектив но борет ся. BOCHS (бес плат ная
вир туаль ная машина со  встро енным отладчи ком)  — единс твен ный разум ный выбор, но,
сколь ко гру зит ся на нем Windows, луч ше не говорить.

    

Крис Касперски
Известный российский хакер. Легенда ][, ex-редактор ВЗЛОМа. Также известен под псевдонимами
мыщъх, nezumi (яп. 鼠, мышь), n2k, elraton, souriz, tikus, muss, farah, jardon, KPNC.

Карманный Arch. Делаем флешку с
живым образом Arch Linux

2 дня назад

Дыры в openSIS. Разбираем SQL-
инъекции на живом примере

4 дня назад

Как дела в контейнере? Собираем
события безопасности в Docker

5 дней назад

HTB Ouija. Повышаем привилегии
через бинарную уязвимость
целочисленного переполнения

6 дней назад

Карманный Arch. Делаем флешку с
живым образом Arch Linux

2 дня назад

Липовый соникс. Реверсим
картридж Liposonix и пишем его
эмулятор

11.04.2024

Новая диета для Linux. Загружаем
современный Linux, используя
минимум памяти

27.03.2024

Фишинг в соцсетях. Как социальные
сети помогают хакерам

20.10.2023

Уязвимость в GitLab позволяла захватывать чужие
аккаунты

Qnap исправила RCE в своих NAS после критики со
стороны исследователей

Эксперты считают, что Apple Wi-Fi Positioning
System (WPS) можно использовать для массовой
слежки

Хакер взломал мошеннический колл-центр и
предупредил пострадавших об обмане

Вымогатель ShrinkLocker использует BitLocker для
шифрования данных

2 дня назад

2 дня назад

2 дня назад

2 дня назад

2 дня назад

